# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Tom Gundersen <teg@jklm.no>
# Contributor: damir <damir@archlinux.org>
# Contributor: Kevin Edmonds <edmondskevin@hotmail.com>

pkgname=libmtp
pkgver=1.1.15
pkgrel=1
pkgdesc="Library implementation of the Media Transfer Protocol"
arch=(x86_64)
url="http://libmtp.sourceforge.net"
license=(LGPL)
depends=(libusb)
source=("https://downloads.sourceforge.net/$pkgname/$pkgname-$pkgver.tar.gz"{,.asc})
sha256sums=('d040900b46757e311b1fb3bfa8c05db09c8b6811e044bce8c88c9f3f6d3a3021'
            'SKIP')
validpgpkeys=(7C4AFD61D8AAE7570796A5172209D6902F969C95) # Marcus Meißner <meissner@suse.de>

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr --with-udev=/usr/lib/udev
  make
}

check() {
  cd $pkgname-$pkgver
  make -k check
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
}
